Datentypen-Schlüsselwörter (Zusammenfassung)

Aktion
Schlüsselwörter

Festlegen integrierter Datentypen.

Boolean, Byte, Currency, Date, Double, Integer, Long, Object, Single, String, Variant (Voreinstellung)

Boolean-Datentyp

Variablen vom Datentyp Boolean werden als 16-Bit-Zahlen (2 Bytes) gespeichert, die nur die Werte True oder False annehmen können. Variablen vom Datentyp Boolean werden als Wahr oder Falsch ausgegeben, wenn Print verwendet wird, bzw. als #TRUE# oder #FALSE#, wenn Write # verwendet wird. Mit den Schlüsselwörtern True und False weisen Sie einer Variablen vom Typ Boolean einen von zwei Zuständen zu.

Beim Umwandeln anderer numerischer Datentypen in Werte des Typs Boolean wird 0 zu False, und alle anderen Werte werden zu True. Beim Umwandeln von Werten des Datentyps Boolean in andere Datentypen wird False zu 0 und True zu -1.

Byte-Datentyp

Variablen vom Datentyp Byte werden als einzelne 8-Bit-Zahlen (1 Byte) ohne Vorzeichen gespeichert und haben einen Wert im Bereich von 0 bis 255.

Der Datentyp Byte ist zur Speicherung binärer Daten nützlich.

Currency-Datentyp

Variablen vom Datentyp Currency werden als 64-Bit-Zahlen (8 Bytes) in einem ganzzahligen Format gespeichert und durch 10.000 dividiert, was eine Festkommazahl mit 15 Vorkomma- und 4 Nachkommastellen ergibt. Diese Darstellung ergibt einen Wertebereich von -922.337.203.685.477,5808 bis 922.337.203.685.477,5807. Das Typkennzeichen für Currency ist das Zeichen (@).

Der Datentyp Currency eignet sich besonders für Berechnungen mit Geldbeträgen und für Festkommaberechnungen, die eine hohe Genauigkeit erfordern.

Date-Datentyp

Variablen vom Datentyp Date werden als 64-Bit-Gleitkommazahlen (8 Bytes) nach IEEE gespeichert und können ein Datum im Bereich vom 01. Januar 100 bis zum 31. Dezember 9999 und eine Uhrzeit im Bereich von 0:00:00 bis 23:59:59 speichern. Jeder gültige Wert eines Datums- oder Zeitliterals kann einer Variablen vom Datentyp Date zugewiesen werden. Ein Datumsliteral muß durch das Zeichen (#) eingeschlossen sein, zum Beispiel: #January 1, 1993# oder #1 Jan 93#.

Variablen vom Datentyp Date verwenden zur Darstellung des Datums das auf dem Computer eingestellte kurze Datumsformat. Zeitangaben werden mit dem auf dem Computer eingestellten Zeitformat (entweder 12- oder 24-Stunden) dargestellt.

Beim Umwandeln anderer numerischer Datentypen in Werte des Datentyps Date repräsentieren die Vorkommastellen das Datum und die Nachkommastellen die Uhrzeit. Mitternacht entspricht dem Wert 0, und Mittag entspricht den Nachkommawert 0,5. Negative ganze Zahlen repräsentieren ein Datum vor dem 30. Dezember 1899.

Double-Datentyp

Variablen vom Datentyp Double (Gleitkommazahl mit doppelter Genauigkeit) werden als 64-Bit-Gleitkommazahlen (8 Bytes) nach IEEE im Bereich von -1,79769313486232E308 bis -4,94065645841247E-324 für negative Werte und von 4,94065645841247E-324 bis 1,79769313486232E308 für positive Werte gespeichert. Das Typkennzeichen für Double ist das Zeichen (#).

Integer-Datentyp

Variablen vom Datentyp Integer werden als 16-Bit-Zahlen (2 Bytes) in einem Bereich von -32.768 bis 32.767 gespeichert. Das Typkennzeichen für Integer ist das Zeichen (%).

Mit Variablen vom Datentyp Integer können Sie auch Aufzählungswerte darstellen. Ein Aufzählungswert besteht aus einer endlichen Menge eindeutiger ganzer Zahlen, von denen jede im verwendeten Kontext eine spezielle Bedeutung hat. Aufzählungswerte erlauben eine einfache Auswahl aus einer bestimmten Anzahl von Möglichkeiten, z.B. 0 = schwarz, 1 = weiß usw. Sie sollten die Const-Anweisung verwenden, um Konstanten für jeden Aufzählungswert zu definieren.

Long-Datentyp

Variablen vom Datentyp Long (lange Ganzzahl) werden als 32-Bit-Zahlen (4 Bytes) mit Vorzeichen im Bereich von -2.147.483.648 bis 2.147.483.647 gespeichert. Das Typkennzeichen für Long ist das Zeichen (&).

Object-Datentyp

Variablen vom Datentyp Object werden als 32-Bit-Adressen (4 Bytes) gespeichert, die auf Objekte in einer Anwendung verweisen. Einer Variablen, die als Object deklariert wurde, kann anschließend mit der Set-Anweisung ein Verweis auf jedes von der Anwendung erzeugte Objekt zugewiesen werden.

Anmerkung   
Obwohl eine Variable, die mit dem Datentyp Object deklariert wurde, einen Verweis auf jedes beliebige Objekt enthalten kann, erfolgt die Verbindung mit dem Objekt, auf das verwiesen wurde, immer zur Laufzeit (Binden zur Laufzeit). Sie können eine frühere Bindung (Binden zur Kompilierungszeit) erzwingen, indem Sie den Objektverweis einer Variable zuweisen, die durch den Namen einer Klasse deklariert wurde.

Single-Datentyp

Variablen vom Datentyp Single (Gleitkommazahl mit einfacher Genauigkeit) werden als 32-Bit-Gleitkommazahlen (4 Bytes) nach IEEE im Bereich von -3,402823E38 bis -1,401298E-45 für negative Werte und von 1,401298E-45 bis 3,402823E38 für positive Werte gespeichert. Das Typkennzeichen für Single ist das Ausrufezeichen (!).

String-Datentyp

Es gibt zwei Arten von Zeichenfolgen: Zeichenfolgen variabler Länge und Zeichenfolgen fester Länge.

· Zeichenfolgen variabler Länge können bis zu 2 Milliarden (oder 2^31) Zeichen enthalten.

· Zeichenfolgen fester Länge können 1 bis etwa 64 KB (2^16) Zeichen enthalten.

Anmerkung   
Zeichenfolgen fester Länge mit dem Attribut Public können in Klassenmodulen nicht verwendet werden.

Die Codes für Zeichen vom Datentyp String liegen im Bereich von 0 bis 255 (einschließlich). Die ersten 128 Zeichen (0 bis 127) entsprechen den Buchstaben und Symbolen auf einer US-amerikanischen Standardtastatur. Diese ersten 128 Zeichen stimmen mit den im ASCII-Zeichensatz definierten Zeichen überein. Die zweiten 128 Zeichen (128 bis 255) sind Sonderzeichen, z.B. Buchstaben aus internationalen Alphabeten, Akzentzeichen, Währungssymbole und Symbole für mathematische Brüche. Das Typkennzeichen für String ist das Dollarzeichen ($).

Datentyp Variant

Interne Darstellung der Werte in Variant-Variablen

Variant-Variablen verfügen über eine interne Darstellung der gespeicherten Werte. Diese interne Darstellung bestimmt, wie Visual Basic die Werte bei Vergleichen und anderen Operationen interpretiert. Wenn Sie einer Variant-Variablen einen Wert zuweisen, verwendet Visual Basic die kompakteste Darstellung, mit der dieser Wert genau gespeichert werden kann. Bei späteren Operationen kann Visual Basic die interne Darstellung einer Variablen ändern. Eine Variant-Variable ist keine Variable ohne Datentyp, sondern eher eine Variable, die ihren Datentyp frei ändern kann. Diese internen Darstellungen entsprechen den expliziten Datentypen, die im Abschnitt „Datentypen“ weiter oben in diesem Kapitel beschrieben werden.

Anmerkung Ein Wert vom Typ Variant nimmt immer 16 Byte ein, unabhängig davon, was Sie darin speichern. Objekte, Zeichenfolgen und Datenfelder werden nicht physisch im Wert vom Typ Variant gespeichert. Für diese Fälle werden 4 Bytes des Wertes vom Typ Variant verwendet, die entweder eine Objektreferenz oder einen Zeiger auf die Zeichenfolge bzw. das Datenfeld enthalten. Die tatsächlichen Daten werden an einer anderen Stelle gespeichert.

In der Regel brauchen Sie sich nicht mit der internen Darstellung der Variablen zu befassen, da Visual Basic die Umwandlungen automatisch durchführt. Falls Sie jedoch wissen möchten, wie eine bestimmte Variable intern dargestellt wird, können Sie die VarType-Funktion verwenden.

Wenn Sie z.B. Werte mit Nachkommastellen in einer Variant-Variablen speichern, verwendet Visual Basic immer die interne Darstellung des Datentyps Double (Gleitkommazahl doppelter Genauigkeit). Wenn Sie wissen, daß Ihre Anwendung die höhere Genauigkeit des Datentyps Double (und die damit verbundene langsamere Ausführung) nicht benötigt, können Sie Ihre Berechnungen beschleunigen, indem Sie die Werte in den Datentyp Single (Gleitkommazahl einfacher Genauigkeit) oder sogar Currency (Währung) umwandeln:

If VarType(X) = 5 Then X = CSng(X)   ' In Single umwandeln.

Bei einer Datenfeldvariablen ist der Wert von VarType die Summe der Rückgabewerte für das Datenfeld und den Datentyp. Das folgende Datenfeld enthält z.B. Double-Werte:

Private Sub Form_Click()

   Dim dblSample(2) As Double

   MsgBox VarType(dblSample)

End Sub

In zukünftigen Versionen von Visual Basic werden möglicherweise weitere Darstellungen für den Datentyp Variant enthalten sein. Deshalb sollte der Code, der aufgrund des Rückgabewerts von VarType Entscheidungen trifft, auch derzeit nicht definierte Werte berücksichtigen und verarbeiten können.

Numerische Werte in Variant-Variablen

Wenn Sie ganze Zahlen in einer Variant-Variablen speichern, verwendet Visual Basic die kleinste mögliche Darstellung. Bei einer kleinen Zahl ohne Nachkommastellen verwendet die Variant-Variable die Darstellung Ganzzahl (Integer). Wenn Sie danach einen größeren Wert zuweisen, ändert Visual Basic die Darstellung in eine lange Ganzzahl (Long) oder, wenn der Wert sehr groß ist oder Nachkommastellen hat, in den Typ Double.

Manchmal soll eine bestimmte Darstellung für eine Zahl verwendet werden. Eine Variant-Variable kann z.B. einen numerischen Wert mit dem Typ Currency speichern, um spätere Rundungsfehler zu vermeiden. In Visual Basic sind eine Reihe von Umwandlungsfunktionen verfügbar, mit denen Sie Werte in einen bestimmten Datentyp umwandeln können (siehe auch „Umwandeln von Datentypen“ weiter oben in diesem Kapitel). Sie können beispielsweise einen Wert in den Typ Currency umwandeln, indem Sie die CCur-Funktion verwenden:

Wochensatz = CCur(Stunden * Stundensatz)

Sie erhalten eine Fehlermeldung, wenn Sie versuchen, eine mathematische Operation oder Funktion auf eine Variant-Variable anzuwenden, die weder eine Zahl noch einen Wert, der als Zahl interpretiert werden kann, enthält. Sie können z.B. keine arithmetischen Operationen mit dem Wert „U2“ ausführen, auch wenn darin eine Zahl enthalten ist, da der Gesamtwert keine gültige Zahl ist. Genauso können Sie keine Berechnungen mit dem Wert „1040EZ“ durchführen. Sie können aber mit Werten wie „+10“ oder „-1,7E6“Berechnungen durchführen, da sie gültige Zahlen darstellen. Aus diesem Grund sollten Sie immer überprüfen, ob eine Variant-Variable einen Wert enthält, der als Zahl interpretiert werden kann. Verwenden Sie dazu die IsNumeric-Funktion:

Do

   EineZahl = InputBox("Geben Sie eine Zahl ein")

Loop Until IsNumeric(EineZahl)

MsgBox "Das Quadrat der Zahl ist: " & (EineZahl*EineZahl)

Wenn Visual Basic nicht-numerische Darstellungen (wie Zeichenfolgen mit einer Zahl) in einen numerischen Wert umwandelt, werden die in den Ländereinstellungen der Windows-Systemsteuerung festgelegten Tausendertrennzeichen, Dezimalzeichen und Währungssymbole verwendet.

Die beiden folgenden Anweisungen ergeben daher den Wert True, wenn als Ländereinstellung in der Systemsteuerung Vereinigte Staaten, Kanada oder Australien definiert ist:

IsNumeric("$100")

IsNumeric("1,560.50")

Die nächsten beiden Anweisungen würden dagegen False ergeben:

IsNumeric("DM100")

IsNumeric("1.560,50")

Genau umgekehrte Ergebnisse (False in den ersten beiden Anweisungen und True in den letzten beiden) erhalten Sie, wenn als Ländereinstellung in der Systemsteuerung Deutschland definiert ist.

Wenn Sie eine Variant-Variable mit einer Zahl einer Zeichenfolgenvariablen oder -Eigenschaft zuweisen, wandelt Visual Basic die Zahlendarstellung automatisch in eine Zeichenfolge um. Mit der CStr-Funktion können Sie Umwandlung von einer Zahl in eine Zeichenfolge explizit durchführen. Sie können aber auch eine Zahl mit der Format-Funktion in eine Zeichenfolge umwandeln, die Formatierungen wie Währungssymbole, Tausendertrennzeichen und Dezimalzeichen enthält. Die Format-Funktion verwendet automatisch die Symbole aus den Ländereinstellungen der Windows-Systemsteuerung.

Zeichenfolgen in Variant-Variablen

Im allgemeinen ergeben sich beim Speichern und Verwenden von Zeichenfolgen in Variant-Variablen wenig Probleme. Allerdings kann das Ergebnis des Operators + mehrdeutig sein, wenn er mit zwei Variant-Variablen verwendet wird. Wenn beide Variant-Variablen Zahlen enthalten, addiert der Operator + die beiden Werte. Enthalten beide Variant-Variablen Zeichenfolgen, dann werden die Zeichenfolgen durch den Operator verkettet. Komplizierter wird die Situation, wenn eine der Variablen als Zahl und die andere als Zeichenfolge dargestellt ist. Visual Basic versucht dann zuerst, die Zeichenfolge in eine Zahl umzuwandeln. Ist die Umwandlung erfolgreich, so addiert der Operator + die beiden Werte. Ist die Umwandlung nicht möglich, erscheint die Fehlermeldung 
Typen unverträglich.

Wenn auf jeden Fall eine Verkettung durchgeführt werden soll, unabhängig von der Darstellung der Werte in der Variablen, sollten Sie den Operator & verwenden. Beispiel:

Sub Form_Click ()

   Dim X, Y

   X = "6"

   Y = "7"

   Print X + Y, X & Y

   X = 6

   Print X + Y, X & Y

End Sub

Dieser Code erzeugt folgendes Ergebnis im Formular:

67      67

13      67

Anmerkung Visual Basic speichert Zeichenfolgen intern als Unicode. Weitere Informationen zum Unicode finden Sie in „Internationale Anwendungen“.

Datums- und Zeitwerte in Variant-Variablen

Variant-Variablen können auch Datums- und Zeitwerte enthalten. Verschiedene Funktionen geben Datums- und Zeitwerte zurück. DateSerial gibt z.B. die Anzahl der verbliebenen Tage im Jahr zurück:

Private Sub Form_Click ()

   Dim Jetzt, Resttage, Reststunden, Restminuten

   Jetzt = Now         ' Now liefert aktuelle(s) Datum/Uhrzeit.

   Resttage = Int(DateSerial(Year(Jetzt) + 1, 1, 1) - Jetzt)

   Reststunden = 24 - Hour(Jetzt)

   Restminuten = 60 - Minute(Jetzt)

   Print Resttage & " Tage verbleiben im Jahr."

   Print Reststunden & " Stunden verbleiben heute."

   Print Restminuten & " Minuten verbleiben in dieser Stunde."

End Sub

Sie können auch mathematische Operationen mit Datums- und Zeitangaben durchführen. Wenn Sie Ganzzahlen addieren oder subtrahieren, werden Tage addiert oder subtrahiert. Addieren oder subtrahieren Sie Brüche, so werden Zeiten addiert oder subtrahiert. Wenn Sie also 20 addieren, werden 20 Tage hinzugezählt, wenn Sie 1/24 subtrahieren, wird eine Stunde abgezogen.

In Variant-Variablen können Sie Datumsangaben im Bereich vom 1. Januar 0100 bis zum 31. Dezember 9999 speichern. Berechnungen mit diesen Daten berücksichtigen keine Kalenderänderungen vor dem Wechsel zum gregorianischen Kalender. Deshalb sind die Berechnungen falsch, die Daten vor dem Jahr ergeben, in dem der gregorianische Kalender eingeführt wurde (1752 in Großbritannien und den zugehörigen Kolonien, im 16. Jh. in vielen anderen Ländern).

Sie können Datums- und Zeitliterale in Ihrem Code verwenden, indem Sie sie so in das Zeichen # einschließen, wie Sie sonst Zeichenfolgenliterale in Anführungszeichen ("") einschließen. Sie können z.B. eine Variant-Variable, die eine Datums-/Zeitangabe enthält, mit einem feststehenden Datum vergleichen:

If EinDatum > #3/6/93# Then

In gleicher Weise können Sie eine Datums-/Zeitangabe mit einem vollständigen Datums-/Zeitliteral vergleichen:

If EinDatum > #3/6/93 1:20pm# Then

Wenn Sie keine Zeit in einem Datums-/Zeitliteral angeben, setzt Visual Basic die Zeitkomponente des Wertes auf Mitternacht (den Anfang des angegebenen Tages). Wenn Sie kein Datum in einem Datums-/Zeitliteral angeben, setzt Visual Basic die Datumskomponente des Wertes auf den 30. Dezember 1899.

Visual Basic erkennt eine Vielzahl von Datums-/Zeitformaten in Datums-/Zeitliteralen. Alle folgenden Beispiele sind zulässig:

EinDatum = #3-6-93 13:20#

EinDatum = #April 27, 1993 1:20am#

EinDatum = #Apr-2-93#

EinDatum = #4 April 1993#

Weitere Informationen Weitere Informationen zur Behandlung internationaler Datumsangaben finden Sie in „Internationale Anwendungen“.

So wie Sie mit der IsNumeric-Funktion ermitteln können, ob eine Variant-Variable einen Wert enthält, der als gültige Zahl interpretiert werden kann, können Sie mit der IsDate-Funktion bestimmen, ob eine Variant-Variable einen zulässigen Datums-/Zeitwert enthält. Dann können Sie den Wert mit der CDate-Funktion in einen Datums-/Zeitwert umwandeln.

Der folgende Code überprüft z.B. die Text-Eigenschaft eines Textfeldes mit IsDate. Wenn die Eigenschaft eine Zeichenfolge enthält, die als gültiges Datum interpretiert werden kann, wandelt der Code den Text in ein Datum um und berechnet die verbleibenden Tage in diesem Jahr:

Dim EinDatum, Resttage

If IsDate(Text1.Text) Then

   EinDatum = CDate(Text1.Text)

   Resttage = DateSerial(Year(EinDatum) + 1, 1, 1) - EinDatum

   Text2.Text = Resttage & " Tage verbleiben in diesem Jahr."

Else

   MsgBox Text1.Text & " ist kein gültiges Datum."

End If

Objekte in Variant-Variablen

Objekte können in Variant-Variablen gespeichert werden. Das kann sehr hilfreich sein, wenn Sie viele verschiedene Datentypen (Objekte eingeschlossen) verarbeiten müssen. Die Elemente eines Datenfeldes müssen z.B. alle den gleichen Datentyp haben. Wenn Sie als Datentyp des Datenfeldes den Typ Variant festlegen, können Sie neben anderen Datentypen auch Objekte im Datenfeld speichern.

